

Tema 3

Nombres reales. Nombres complejos

Objectius

1. Càlculs amb nombres reals.
2. Propietats dels nombres reals.
3. Càlculs amb nombres complexos.
4. Propietats dels nombres complexos.

Contingut

- 03-1. Nombres reals.
- 03-2. Nombres complexos.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.
- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.

- ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable
- JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J.
RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

03-1.- Nombres reals.

Els càlculs desenvolupats en aquest apartat es troben al fitxer **Tema_03-1.wxm**.

El conjunt $\mathbb{R}$ dels nombres reals, dotat d'estructura algebraica de cos commutatiu amb la suma i el producte ordinaris, constitueix el conjunt numèric bàsic per a l'Anàlisi Matemàtica de funcions de variable real. Les operacions internes de suma i producte de nombres reals així com les seves propietats, se suposen conegudes. La sintaxi d'aquestes operacions s'ha vist a la Pràctica 01.

Se suposa coneguda també la ordenació dels nombres reals amb una relació d'ordre total, així com la compatibilitat de les operacions internes de $\mathbb{R}$ amb aquesta relació d'ordre. Aquesta relació d'ordre permet definir els intervals en la recta real:

Interval tancat d'origen a i extrem b : $[a, b] = \{x \in \mathbb{R} : a \leq x \leq b\}$

Interval obert d'origen a i extrem b : $]a, b[= \{x \in \mathbb{R} : a < x < b\}$

L'interval obert es designa també (a, b) ; en aquest Cours aquesta notació es fa servir només per indicar una parella ordenada de nombres. Poden definir-se, de manera evident, intervals de tipus mixt.

Se suposa coneguda també la definició de la recta real ampliada, que resulta d'incloure els símbols $+\infty$ i $-\infty$; així mateix, se suposen conegudes les propietats inherents a aquesta definició, en particular, les regles operatives relatives als símbols $+\infty$ i $-\infty$.

Un concepte important és el de valor absolut d'un nombre real:

$$|x| = \begin{cases} x, & \text{si } x \geq 0 \\ -x, & \text{si } x < 0 \end{cases}$$

Se suposen conegudes les propietats del valor absolut i que aquest valor absolut permet definir una estructura d'espai mètric mitjançant la distància euclidiana ordinària:

$$d(x, y) = |x - y|, \quad x, y \in \mathbb{R},$$

Les propietats d'aquesta distància se suposen també conegudes. En aquesta pràctica es pretén consolidar la sintaxi d'aquests elements, la definició de subconjunts de nombres reals mitjançant equacions o inequacions en les quals intervé el valor absolut i la seva expressió mitjançant intervals. Tots aquests conceptes es veuen als Exemples que segueixen.

Exemple 03-1.- Determineu l'interval de la recta real que correspon al subconjunt de nombres reals definit per

$$A = \{x \in \mathbb{R} : x^2 + x - 6 < 0\}.$$

Resolució. Dissenyem per $P(x) = x^2 + x - 6$. En primer lloc cal determinar els punts en que es compleix la condició $P(x) = 0$:

```
(%i1) solve(x^2+x-6=0, x);
(%o1) [x = -3, x = 2]
```

Ara cal analitzar el signe del trinomi $P(x) = x^2 + x - 6$ en cada un dels tres intervals $]-\infty, -3[,]-3, 2[$ i $]2, +\infty[$:

```
(%i2) assume(x<-3);
      is(x^2+x-6<0);
(%o2) [x < -3]
(%o3) false
(%i4) forget(x<-3);
      assume(x>-3, x<2);
      is(x^2+x-6<0);
(%o4) [x < -3]
(%o5) [x > -3, x < 2]
(%o6) true
(%i7) forget(x>-3, x<2);
      assume(x>2);
      is(x^2+x-6<0);
(%o7) [x > -3, x < 2]
(%o8) [x > 2]
(%o9) false
```

En conclusió, es compleix: $A =]-3, 2[$.

Exemple 03-2.- Determineu l'interval de la recta real que correspon al subconjunt de nombres reals definit per:

$$A = \left\{x \in \mathbb{R} : \frac{2x+1}{x+2} < 1\right\}.$$

Resolució. En primer lloc veiem que la condició es pot expressar de forma equivalent mitjançant:

$$\frac{2x+1}{x+2} < 1 \iff \frac{2x+1}{x+2} - 1 < 0 \iff \frac{2x+1-(x+2)}{x+2} = \frac{x-1}{x+2} < 0$$

Per tant, treballarem amb aquesta formulació equivalent més senzilla. En referència al fitxer Tema02-1.wxm, observeu que està estructurat indicant els exemples amb una cel·la de text; és convenient començar cada nou apartat de càlculs amb l'anul·lació d'assignacions anteriors fent servir la instrucció “kill(all)”.

Establerta doncs la condició, ara cal verificar si es compleix o no aquesta en cada un dels tres intervals $]-\infty, -2[$, $]-2, 1[$, $]1, +\infty[$:

```
(%i1) assume(x<-2);
      is((x-1)/(x+2)<0);
(%o1) [x < -2]
      (%o2) false
(%i3) forget(x<-2)$
      assume(x>-2, x<1);
      is((x-1)/(x+2)<0);
(%o4) [x > -2, x < 1]
      (%o5) true
(%i6) forget(x>-2, x<1)$
      assume(x>1);
      is((x-1)/(x+2)<0);
(%o7) [x > 1]
      (%o8) false
```

Això permet concloure que la solució és $A=]-2, 1[$.

Exemple 03-3.- Si a és un nombre real estrictament positiu, demostreu que es compleix l'equivalència:

$$|x| < a \Leftrightarrow -a < x < a.$$

Resolució. En primer lloc introduïrem la condició de positivitat:

```
(%i1) assume(a>0);
(%o1) [a > 0]
```

Ara verificarem la condició per a les dues possibilitats en relació al nombre x , en primer lloc suposant que és positiu i menor o major que a i en segon lloc, que és negatiu i menor o major que $-a$, és a dir:

$$\begin{aligned} x > 0, x < a; \quad x > 0, x > a \\ x < 0, x > -a; \quad x < 0, x < -a \end{aligned}$$

Per tant, introduint les condicions a wxMaxima resulta:

```

(%i2) assume(x>0, x<a);
      is(abs(x)<a);
(%o2) [x > 0, a > x]
(%o3) true
(%i4) forget(x>0, x<a)$
      assume(x>0, x>a);
      is(abs(x)<a);
(%o5) [x > 0, x > a]
(%o6) false
(%i7) forget(x>0, x>a)$
      assume(x<0, x>-a);
      is(abs(x)<a);
(%o8) [x < 0, x + a > 0]
(%o9) true
(%i10) forget(x<0, x>-a)$
      assume(x<0, x<-a);
      is(abs(x)<a);
(%o11) [x < 0, x + a < 0]
(%o12) false

```

Per tant, l'equivalència queda provada.

Exemple 03-4.- Determineu l'interval de la recta real que correspon al subconjunt de nombres reals definit per:

$$A = \{x \in \mathbb{R} : |x - 2| < 1\}.$$

Resolució. Tenint en compte l'exemple anterior, es pot afirmar que $-1 < (x-2) < 1$, amb la qual cosa es conclou que $A =]1, 3[$. Redacteu com exercici la metodologia de resolució sense tenir en compte l'exemple anterior.

Exemple 03-5.- Determineu l'interval de la recta real que correspon al subconjunt de nombres reals definit per:

$$A = \{x \in \mathbb{R} : |x - 4| - |x - 1| < 0\}.$$

Resolució. En primer lloc verificarem si es compleix la condició a l'esquerra del valor menor dels dos valors de referència, que són $a=1$, $b=4$:

```

(%i1) assume(x<1);
      is(abs(x-4)-abs(x-1)<0);
(%o1) [x < 1]
(%o2) false

```

Ja sabem que en aquest cas no es compleix. Ara analitzem què passa entre els dos valors de referència:

```
(%i3) forget(x<1)$  
      assume(x>1, x<4);  
      is(abs(x-4)-abs(x-1)<0);  
(%o4) [x > 1, x < 4]  
(%o5) unknown
```

La resposta del programa significa que no pot decidir sobre la pregunta formulada, ja que la condició en una part de l'interval és certa i en l'altra no; dividirem l'interval en dues parts iguals i veurem què passa en cada una d'elles:

```
(%i6) forget(x>1, x<4)$  
      assume(x>1, x<5/2);  
      is(abs(x-4)-abs(x-1)<0);  
(%o7) [x > 1, x < 5/2]  
(%o8) false  
(%i9) forget(x>1, x<5/2)$  
      assume(x>5/2, x<4);  
      is(abs(x-4)-abs(x-1)<0);  
(%o10) [x > 5/2, x < 4]  
(%o11) true
```

Finalment, analitzarem què passa a la dreta del major dels dos valors:

```
(%i12) forget(x>1, x<4)$  
      assume(x>4);  
      is(abs(x-4)-abs(x-1)<0);  
(%o13) [x > 4]  
(%o14) true
```

Per tant, la solució és $A =]5/2, +\infty[$.

03-2.- Nombres complexos

Els càlculs desenvolupats en aquest apartat es troben al fitxer **Tema_03-2.wxm**.

El conjunt dels nombres complexos es defineix com el conjunt dels parells ordenats de nombres reals, és a dir:

$$\mathbb{C} = \{(x, y) : x, y \in \mathbb{R}\}$$

Si un nombre complex s'escriu $z = (x, y)$ i aquesta representació s'anomena *forma cartesiana* del nombre complex. El primer nombre real de la parella s'anomena “part real de z ” i s'escriu $x = \text{Re}(z)$; el segon nombre real del parell s'anomena “part imaginària de z ” i s'escriu $y = \text{Im}(z)$.

Aquesta forma o representació no és la que s'utilitza a wxMaxima per representar els complexos: es fa servir l'anomenada forma binòmica que utilitza la unitat imaginària $i = (0, 1)$, de manera que aleshores s'escriu $z = x + yi$. La representació en wxMaxima de la unitat imaginària és %i. Per a major comoditat és convenient fer servir variables per designar els nombres complexos; a més, si es vol que en la sortida quedi explicitat el símbol atorgat a un complex, cal escriure-ho amb la sintaxi que es pot veure tot seguit:

```
(%i1) 'z:=x + y* %i;  
      'Re(z)=realpart(z);  
      'Im(z)=imagpart(z);  
(%o1) z = %i y + x  
(%o2) Re(%i y + x) = x  
(%o3) Im(%i y + x) = y
```

En el conjunt dels amb complexos es defineixen dues operacions internes, suma i producte, que es poden calcular fàcilment amb wxMaxima. La sintaxi de la suma i el producte s'il·lustra a continuació; observeu que, si no s'indica explícitament que es desenvolupi, el producte s'efectua en forma simbòlica, és a dir, es deixa indicat com a tal. En wxMaxima cal aplicar la instrucció “rectform” per explicitar que es vol efectuat el resultat i expressat en forma binòmica. Així doncs:

```
(%i4) 'z1:=a+b*%i;      'z2:=c+d*%i;  
      '(z1+z2)=z1+z2;    '(z1*z2)=z1*z2;    '(z1*z2)=rectform(z1*z2);  
(%o4) z1 = %i b + a  
(%o5) z2 = %i d + c  
(%o6) z2 + z1 = %i d + c + %i b + a  
(%o7) z1 z2 = (%i b + a) (%i d + c)  
(%o8) z1 z2 = %i (a d + b c) - b d + a c
```

```
(%i9) 'z3=z3: 2.432+3.789*%i;      'z4=z4: 5.321+6.987*%i;
      '(z3+z4)=z3+z4;              '(z3-z4)=z3-z4;
      '(z3*z4)=rectform(z3*z4);
(%o9) z3 = 3.789 %i + 2.432
(%o10) z4 = 6.987 %i + 5.321
(%o11) z4 + z3 = 10.776 %i + 7.753
(%o12) z3 - z4 = -3.198 %i - 2.889
(%o13) z3 z4 = 37.153653000000001 %i - 13.533071
```

La sintaxi per efectuar el quocient entre dos complexos, és a dir, el producte d'un complex per l'invers d'un altre complex respecte del producte, és senzilla. Cal tenir present que també s'ha d'especificar amb la instrucció “rectform” que el quocient es vol desenvolupat, ja que en cas contrari només s'efectua en forma simbòlica i es deixa indicat. Vegem la sintaxi del quocient de complexos:

```
(%i14) 1/z2;      z1*(1/z2);      '(z1/z2)=z1/z2;
(%o14)  $\frac{1}{\%i d + c}$ 
(%o15)  $\frac{\%i b + a}{\%i d + c}$ 
(%o16)  $\frac{z1}{z2} = \frac{\%i b + a}{\%i d + c}$ 
(%i17) '(z1/z2)=rectform(z1/z2);
(%o17)  $\frac{z1}{z2} = \frac{b d + a c}{d^2 + c^2} + \frac{\%i (b c - a d)}{d^2 + c^2}$ 
(%i18) '(z3/z4)=rectform(z3/z4);
(%o18)  $\frac{z3}{z4} = 0.041084341863689 \%i + 0.51100475410667$ 
```

El conjugat d'un nombre complex s'obté amb la instrucció “conjugate”:

```
(%i19) conjugate(z1);
(%o19) a - %i b
```

El mòdul d'un nombre complex $z = a + bi$ s'obté amb la instrucció “abs”:

```
(%i20) abs(z1);      (abs(z1))^2;
(%o20)  $\sqrt{b^2 + a^2}$ 
(%o21)  $b^2 + a^2$ 
```

El producte d'un complex pel seu conjugat dona com a resultat el quadrat del mòdul:

```
(%i22) rectform(z1*conjugate(z1));
(%o22) b2 + a2
```

Si es considera el nombre complex $z = a + bi$, es defineix l'argument d'aquest nombre complex, notat $\arg(z)$, mitjançant:

$$\sin(\arg(z)) = \frac{\operatorname{Im}(z)}{|z|}; \quad \cos(\arg(z)) = \frac{\operatorname{Re}(z)}{|z|}; \quad \tan(\arg(z)) = \frac{\operatorname{Im}(z)}{\operatorname{Re}(z)}$$

En wxMaxima l'argument d'un complex s'obté amb la funció notada “atan2” que té dos arguments i calcula l'angle que té per tangent el quocient del primer entre el segon. La sintaxi és “atan2(b,a)”, és a dir, escrivint primer la part imaginària i després la part real. La instrucció s'aplica fins i tot en el cas en què $a = 0$. Per exemple:

```
(%i23) atan2(sqrt(3),-1);      atan2(1,0);      atan2(0,1);
      atan2(-1,0);      atan2(0,-1);
(%o23)  $\frac{2\pi}{3}$ 
(%o24)  $\frac{\pi}{2}$ 
(%o25)  $-\frac{\pi}{2}$ 
(%o26)  $\pi$ 
(%o27) 0
```

Cal recordar que l'argument d'un complex és un conjunt d'angles i per tant:

$$\arg(-1 + i\sqrt{3}) = \frac{2\pi}{3} + 2k\pi, \quad k \in \mathbb{Z}.$$

L'argument principal correspon a $k = 0$.

La representació exponencial d'un complex s'obté mitjançant:

$$z = \operatorname{Re}(z) + i \operatorname{Im}(z) = |z| \cos(\arg(z)) + i |z| \sin(\arg(z)) = |z| e^{i \arg(z)}$$

En wxMaxima s'obté aquesta forma exponencial amb la instrucció “polarform”:

```
(%i28) polarform(z1);
(%o28)  $\sqrt{b^2 + a^2} e^{i \operatorname{atan2}(b, a)}$ 
```

```
(%i29) z5:(-1)+(sqrt(3))*%i;
      polarform(z5);
(%o29)  $\sqrt{3} \, %i - 1$ 
       $2 \, %e^{\frac{2 \, %i \pi}{3}}$ 
(%i31) polarform(z3);
      polarform(z4);
(%o31) 4.502348831443428 %e1.000166135924386 %i
(%o32) 8.782437588733552 %e0.91993956282176 %i
```

Les potències de nombres complexos es poden calcular fàcilment en forma polar o exponencial, ja que si $z = a + bi = |z|e^{i\theta}$ i $n \in \mathbb{N}$ aleshores:

$$(z)^n = (|z|e^{i\theta})^n = |z|^n e^{in\theta}$$

El programa wxMaxima efectua estrictament aquesta operació:

```
(%i33) (polarform(z1))^n;
(%o33) (b2+a2)n/2 %e%i atan2(b,a) n
```

Així, per exemple:

```
(%i34) (z3)^3;      polarform((z3)^3);
(%o34) (3.789 %i+2.432)3
(%o35) 91.26776600277026 %e3.000498407773157 %i
```

Ara es pot obtenir aquesta potència en la forma binòmica:

```
(%i36) rectform(polarform((z3)^3));
(%o36) 12.83467293899997 %i-90.36081164799998
```

Per suposat aquest resultat es pot obtenir directament calculant la potència en forma binòmica:

```
(%i37) expand(z3^3);
(%o37) 12.834672939 %i-90.36081164800001
```

També es poden dur a terme operacions amb potències de complexos:

```
(%i38) u1:(z3)^2;      u2:(z4)^4;      u3=expand(u1*u2);
(%o38) (3.789 %i+2.432)^2
(%o39) (6.987 %i+5.321)^4
(%o40) u3 = 99322.25822814016-68402.22011408709 %i
(%i41) ((z3)^5*(z4)^6)/(z3+z4)^3;
(%o41) 
$$\frac{(3.789\%i+2.432)^5(6.987\%i+5.321)^6}{(10.776\%i+7.753)^3}$$

(%i42) rectform(%);
(%o42) 357344.1360816646 %i+63138.19216269674
```

Es poden calcular arrels n -èsimes de complexos com potències d'exponent $1/n$.
 Recordeu que si

$$z = a + bi = |z|e^{i\theta},$$

aleshores les arrels n -èsimes d'aquest complex són $u_0, u_1, \dots, u_{n-1}$, i es verifica que:

$$|u_k| = (|z|)^{1/n}; \quad \arg(u_k) = \frac{\theta}{n} + k \frac{2\pi}{n}, \quad k = 0, 1, \dots, n-1$$

Cal tenir en compte, però, que wxMaxima només dona com a resultat una d'aquestes arrels, que és u_0 i si es volen calcular la resta d'arrels cal construir-les explícitament. Així, si es volen calcular les tres arrels cúbiques del complex $z = 1 - i$, aleshores els càlculs amb wxMaxima donen com a resultat:

```
(%i1) z10:1-%i;      abs(z10);
      atan2(-1,1);    polarform(z10);
(%o1) 1-%i
(%o2)  $\sqrt{2}$ 
(%o3)  $-\frac{\pi}{4}$ 
(%o4)  $\sqrt{2} e^{-\frac{\%i \pi}{4}}$ 
(%i5) u0:(z10)^(1/3);      polarform(u0);
(%o5) (1-%i)^(1/3)
(%o6)  $2^{1/6} e^{-\frac{\%i \pi}{12}}$ 
(%i7) theta_1:(- %pi/12)+(2*%pi)/3;
(%o7)  $\frac{7\pi}{12}$ 
```

```
(%i8) u1:(2^(1/6))*exp(%i*7*%pi/12);
(%o8)  $2^{1/6} e^{\frac{7\pi i}{12}}$ 
(%i9) theta_2:(-%pi/12)+(4*%pi)/3;
(%o9)  $\frac{5\pi}{4}$ 
(%i10) u2:(2^(1/6))*exp(%i*5*%pi/4);
      polarform(u2);
(%o10)  $2^{1/6} \left( -\frac{i}{\sqrt{2}} - \frac{1}{\sqrt{2}} \right)$ 
(%o11)  $2^{1/6} e^{-\frac{3\pi i}{4}}$ 
```

Ara es poden expressar les tres arrels cúbiques en forma binòmica:

```
(%i12) rectform(u0),numer;      rectform(u1),numer;
      rectform(u2),numer;
(%o12) 1.084215081491351-0.29051455550725 %i
(%o13) 1.084215081491351 %i-0.29051455550725
(%o14) -0.7937005259841 %i-0.7937005259841
```

Finalment, es pot verificar que s'han determinat les arrels buscades, calculant el cub de cada una de les arrels cúbiques i comprovant que el resultat és el nombre complex inicial. En efecte:

```
(%i15) polarform(u0^3);
      polarform(u1^3);
      polarform(u2^3);
(%o15)  $\sqrt{2} e^{\frac{\pi i}{4}}$ 
(%o16)  $\sqrt{2} e^{\frac{\pi i}{4}}$ 
(%o17)  $\sqrt{2} e^{\frac{\pi i}{4}}$ 
```